
Overview

Visual FoxPro has a number of misleading error messages, probably more than you realize.  Even some of the error messages you’re familiar with can pop up in unexpected places.  This session will cover a fair number of these error messages, will show you what they really mean, and how to get rid of them.  In the process, you’ll learn some debugging techniques to help you in your general programming.

Versions and Resources

Please note that  in versions of VFP prior to version 6, the causes for many of these errors are different from the ones I discuss here.  Therefore, , I won’t discuss anything that only occurs in versions before VFP 6 SP3.

There are two excellent sources for helping you understand errors or any other issues you run into with VFP.  One source is the FoxPro Wiki, run by Steven Black (of INTL fame).  Go to www.fox.wikis.com. The other source is the Microsoft knowledge base (http://support.microsoft.com. Click the link for “Search the Knowledge Base”).  If you enter “Visual FoxPro” (without the quotes) plus the exact error message into the search box, you’ll often find a list of articles referenced.  Traditionally, using  www.google.com has been the best way to search for pages on the Microsoft site, but periodically Microsoft rearranges everything, and it takes Google a little while to record all the changes.  With the changes made to the site this year, I’ve find the Microsoft search feature to be fairly useful.

Under the heading for most messages, I list where you can find an example of the error.  The programs and forms mentioned are available as downloads, and are meant to accompany this paper.

Now, let’s move on to the error messages…

"'field' phrase is not found" (1130)

Examples of this error: Do Form FieldPhrase1 and Do Form FieldPhrase2

This is one of the more poorly worded error messages in VFP.  What’s worse is when you look up this error in the online help, this is what it tells you:

The field specified in PROMPT field was not found. 

· You are using a command with a PROMPT clause that references an unknown field.

You probably weren’t doing anything at all with a PROMPT command when you got this error, and you’re really scratching your head wondering what on earth this means.  This error should have been worded “RowSource is invalid”, because that’s what it’s really trying to tell you. 

If you happened to search the knowledge base instead, you probably found article Q133024, which does help you solve this problem.  That article says:

This is caused by incorrect syntax in the list of fields in the RowSource property. The documentation and online Help imply that the correct way to list fields in this property is: 

   Table1.field1, Table1.field2, Table1.field3, ...

This is incorrect. The correct syntax is: 

   Table1.field1, field2, field3, ...

Any deviation from this syntax (using an expression instead of a field name, trying to use fields from multiple or related tables, and so on) results in the same error message.

Keep in mind that any typo in the field name list causes this error to occur.  And while we’re talking about field names, you should know that just specifying the alias without explicitly listing each field doesn’t work either.

Another cause of this error is using a RowSource of “2 – Alias” for a combobox or listbox when the alias isn’t open in the current data session.

I’ve even seen this error occur when I had left the RowSourceType as “3 – SQL Statement”, but changed the RowSource to an array.

"Cannot access the selected table" (1152)

Example of this error: Do Form CannotAccess

This one is symptomatic of closing the table used for the RowSource for a combobox or listbox too early.  If you are using a form’s data environment to close tables for you, you don’t have to worry about the tables being closed too early.  If you’re closing tables yourself, make sure you close them in the Destroy method of the form.  I like to have this code in my base form class Destroy method:

IF This.DataEnvironment.AutoCloseTables = .F.


CLOSE DATABASES ALL
&& Close any DBC in this datasession


CLOSE TABLES

&& Close any free tables

ENDIF

You can also get this error if you have more than one combobox or listbox on your form where the RowSource is a SQL Select statement, and you use the same cursor as the output more than once.

“Record is in use by another” (109)

Besides the times when you’d expect to get this error--namely in a multi-user environment when someone has a record locked and another user tries to lock the same record--it’s possible to get this error on a form with a grid when you’re the only one running the system. This error can occur when you’re trying to run a REPLACE or TABLEUPDATE command if the focus is still on the grid.  To fix this problem, simply SetFocus to another object on the form before executing these commands, then SetFocus back to the grid afterwards.  If there is no other object on the form for you to SetFocus to, then you need to create an object for this purpose.  You can set the Visible property of this object to .F. so that the user never sees it.

This error should not be confused with “Update conflict” which you can also get while you’re the only person running the system.  This error is common if you’re using a view and are trying to run a TABLEUPDATE command, but there is something wrong with your view definition.

“Variable ‘XXXX’ is not found”/“File ‘XXXX’ does not exist” (12/1)

Example of this error: Do Form VariableNotFound

Here’s a good, short example that demonstrates this error:

ON ERROR do ErrHandler

? lcName

ON ERROR

*---------------------------------------------------------------

PROCEDURE ErrHandler

AERROR(laError)

DISPLAY MEMORY LIKE laError

=VARTYPE(unknown)

AERROR(laError)

DISPLAY MEMORY LIKE laError

ENDPROC

What happens here is that something triggers your error handler (in this case, the display of an unknown variable), and your error handler then uses the TYPE or VARTYPE function on an unknown variable.  Use of either of these functions with an unknown variable changes the error information that AERROR stores.  The key to avoiding this problem is to call the AERROR function before using TYPE or VARTYPE.
 

Incidentally, the example above produces the “Variable ‘XXXX’ not found” flavor of this message because the syntax of the line that triggered the call to the error handler indicated we’re looking for a variable.  If you change ? lcName to something like USE XYZ, you’ll get the “File ‘XXXX’ not found” flavor instead.

Another cause for this is when your app crashes during testing, which leaves the app's SYSMENU in place. Variables or object references used for SKIP FOR clauses often go out of scope.

"’Cursor’ must be created with SELECT…INTO TABLE" (1815)

Example of this error: Do Form CursorMustBe

This error occurs when an SQL statement is resolved by creating a filtered view of the source table instead of a physical cursor. Subsequent attempts to access this cursor by using a SQL SELECT causes this message to be displayed. A filtered view of the source table is produced when: 

1. The query is fully optimized

2. Only a single table is used as a source

3. There are no constants used in the field list

You can tell if the result is a filtered view or has a real disk file by using the DBF() command.  If the result is the name of the source table instead of a .TMP file, then it’s a filtered view. Starting with VFP 6, there is a NOFILTER clause that will force VFP to create a physical  file for the cursor instead of a filtered view.  The READWRITE clause, added in VFP 7, will also force VFP to create a  file on disk for the cursor.

"Feature is not available" (1001)

Example of this error: Run FeatureNotAvai.exe in the runtime environment

This error indicates that you are trying to use a command in a runtime environment that is only available in the development environment.  Commands that cause this error are listed below.

SET STEP ON


MODIFY QUERY

SUSPEND


MODIFY SCREEN

MODIFY CONNECTION

CREATE SCREEN

MODIFY DATABASE

MODIFY VIEW

MODIFY FORM


CREATE FORM

MODIFY MENU


CREATE MENU

MODIFY PROCEDURE

CREATE QUERY

MODIFY PROJECT 

CREATE VIEW (if you have a DBC open, otherwise, you get a VUE)

Note: Starting with VFP 7, SET STEP ON is ignored in executables (EXE, APP, DLL).

In addition, you can get a “Feature not available” error in the development environment if you issue a COMRETURNERROR() command from the Command Window or from a program running in development, but not running as a COM component.
  If you call this command from a non-COM object in a runtime environment, you won’t get the “Feature not available” error, but your app will just terminate at that point.

“Cannot create program workspace” (***)

This indicates a problem with creation of the temporary files that VFP needs while running.  The error occurs when either the location specified by the TMPFILES system variable does not exist or the current user has insufficient rights to it.  The current user must have read/write/delete permissions for the temp files directory.

You can specify where these files are created either through the Tools | Options menu, on the File Locations page, or you can specify a setting for TMPFILES in a config.fpw file.  If you do both, the config.fpw file takes precedence.  If you do neither, what happens depends on whether you’re running in development or a runtime environment.  In development, the temp files will go into the directory where VFP is installed.  In the runtime environment, if you have a Temp environment variable that points to a valid directory, VFP puts the temp files there, otherwise, they go into the directory where your app is running.

Note: Although I have had this problem under Windows NT, I’ve been unable to duplicate it in Windows 2000.  If you don’t have the required permissions to the specified temp directory, Windows 2000 will just locate the temporary files elsewhere, usually in the Windows temp directory.

"Cannot locate the Microsoft Visual FoxPro support Library"

Example of this error: The program CannotLocate.exe was compiled in VFP 5.  Run this program on a machine that has never had the VFP 5 runtime installed.

This error indicates one or more of the required runtime files can’t be found or are not in the registry.  The exact file names vary with the version of Visual FoxPro.  The basic files required for VFP 6, 7 and 8 are listed in figure 1.  Note that there may be other required files, based on components used in your application.

	VFP 8
	VFP 7
	VFP 6

	VFP8r.dll
	VFP7r.dll
	VFPr.dll

	VFP8t.dll (MTDLL support)
	VFP7t.dll (MTDLL support)
	VFP6t.dll (MTDLL support)

	VFP8renu.dll (for English)
	VFP7renu.dll
	VFP6renu.dll (for English)

	MSVCR70.dll
	MSVCR70.dll
	


Figure 1.  Required runtime files.

In many cases, these files can be placed in the same directory with your app, and don’t have to be registered on the server or each workstation.  If you don’t place them in the directory with the app, then you must register them.

If you have HTML help called from your system, you also need to install the HTML engine files.  These are FoxhhelpX.exe and FoxhhelppsX.dll where X is the VFP version.  These files can also go into the directory with your app, even if you have a situation where the other runtime files can’t.

“The VFPXR.DLL file is invalid or damaged”

Example of this error: The program vfp6rinvalid.exe was compiled in VFP 6.  Make sure you’ve installed the VFP 6 runtimes, then erase the files or remove all access rights to the runtime files.

This error is very similar to “Cannot locate the Microsoft Visual FoxPro support library”.  The difference is, if you get this error, your runtime files are registered, but the files can’t be accessed.  Check permissions to the directory where the runtime files are located, and to the files themselves.  It may be necessary  to unregister each of the runtime files and re-register them, using the Windows REGSVR32 command.

"Function name is missing )" (1300)

Example of this error: Do Form FunctionMissing

This can be a deceptive error message because it’s not always a case of a missing parenthesis. As an example look at this code: 

IIF(INLIST(UPPER(POClass)"RENTAL","BLANKET"),0, POAmtTax-PaidAmt)

Note that a comma is missing after UPPER(POClass).  This code causes a “Function name is missing )” error, because of the missing comma.

Here’s another example:

ASSERT  !("ACTIVATE" & Program(Program(-1)-1) AND ;


This.txtSum.Value # This.nTarget) ;


MESSAGE Program() + ;

" target and sum of split are different on entry."

The problem in the above code is that it’s using an ampersand (&) instead of a dollar sign ($), but it will yield the “Function name is missing )” error.

The lesson here is to not waste too much time looking for a missing parenthesis if it appears they are all properly matched.  The answer is likely some other problem.

Another situation that’s likely to cause this error is trying to use a MIN() or MAX() function in an SQL Select statement.  The problem is that when you’re in an SQL Select, VFP doesn’t use it’s own MIN/MAX function, but the SQL aggregate functions MIN and MAX, which have different syntax and different usage.  If you need a VFP-style MIN or MAX function in your SQL Select, you must use a IIF() function instead—this is assuming you’re querying native data and not a client/server backend. For example, this code causes a “Function name is missing )” error:

SELECT MAX(Val1, Val2) ;

   FROM DemoData ;

   INTO CURSOR Temp

The query should be constructed like this instead:

SELECT IIF(Val1 > Val2, Val1, Val2) ;

   FROM DemoData ;

   INTO CURSOR Temp

“File is not open” (1113)

This error occurs when there’s a problem with one of your index tags.  Typically, you’ve created a tag that includes a field in a foreign table.  The easiest way to fix this is to delete your CDX and rebuild your index tags.

“Data type is invalid for this property (1732)”

Example of this error: Do Form DataTypeInvalid

Most of the time you get this error, it’s because you’re trying to do something like set a logical field to a character value.  Sometimes, though, you’ll get this error when you’re trying to set a character field to a character expression, and you don’t understand why you’re getting an error.  If you display the expression in the watch window, it shows the type as a character.  The problem is likely that one of the values in your expression is a null value.  For example, if you try to set the caption of a label to something like this:

Thisform.lblSomething.Caption = “Specify price per “+UnitAbbrev

If the UnitAbbrev in this case is null, you’ll get this error.  Make sure you use the NVL function in any situation where there may be a null.

Thisform.lblSomething.Caption = “Specify price per “+NVL(UnitAbbrev,’’)

 “Nesting error” (96)

Example of this error: Do NestingError

I’ve seen a couple of causes of the “Nesting error”.  The most common is when you use an EXIT or LOOP command when you’re not in a loop. This is easy to do if you copy code from a loop to a location outside of a loop.  The solution is obviously to remove the offending code.

The second case is harder to track down. I had this happen to me on a DO WHILE line of code that was checking a property in a COM object.  There was a syntax error in the code checking the property, but instead of a more obvious error message, I got a “Nesting error”.

“An IF | ELSE | ENDIF statement is missing” (1211)

Example of this error: Do Form IfElseEndif

At first you’d expect fixing this kind of error to be quite easy.  After all, VFP shows you the line of code where you’re missing the statement, right?  Wrong!  The line that VFP points to can be way off base.  Consider the code shown below.  I’ve included the line numbers so  specific lines can be referenced easily.

1  IF llSomeExpr

2     IF TestSomeCondition

3       * Do all sorts of code here.

4     * Got some other code here.

5     IF SomeThingWorks

6       * Do lots of stuff here.

7     ENDIF 

8  ELSE

9     * All the above stuff doesn't happen

10    * I want to do something completely different

11    * Blah

12    * Blah

13    * Blah

14 ENDIF

If you compile the code above, it will tell you the error is on line 14.  That’s the last line of the code.  If you look closely, however, you’ll notice that there’s an ENDIF missing between lines 4 and 5.  If you have a really hairy set of code with lots of nested IF | ELSE | ENDIF code, this can become quite difficult to debug!  One thing that can help in cases where the code is short is to run Beautify on your code.  The Beautify option can be found on the Tools menu, and will be enabled while you’re editing the program.  Make sure you make a backup copy of your program first.  Beautify will indent where it believes there should be indents.  Comparing the output of Beautify with your original code and seeing where the indentations differ is one way to find the error.

If your code also has other looping and/or branching constructs involved, one tool I find even more helpful than Beautify is the Action Diagram created by the Documentation Wizard. For example, the code below will generate a whole slew of problems, starting with a “There is a missing keyword in the FOR...ENDFOR or DO CASE...ENDCASE command structure” error on line 17, which is an OTHERWISE command.  Looking over the DO CASE…ENDCASE structure, there doesn’t appear to be any problem with the code.

 1 IF llSomeExpr

 2    DO WHILE .T.

 3       DO CASE

 4       CASE .F.

 5          * Some code goes here

 6          * Do some other stuff here

 7       CASE .F.

 8          * Blah

 9          * Blah

10          * Blah

11          IF TestSomeCondition

12             * Do all sorts of code here

13          ELSE

14          IF SomeThingWorks

15             * Do lots of stuff here.

16          ENDIF 

17       OTHERWISE

18          * Do some stuff here to trap for all other cases

19          DO CASE

20          CASE Othertests

21             * All sorts of code here

22             * that makes this

23             * case long

24          CASE StillOtherTests

25          ENDCASE

26       ENDCASE 

27    ENDDO

28 ELSE

29    * All the above stuff doesn't happen

30    * I want to do something completely different

31    * Blah

32    * Blah

33    * Blah

34 ENDIF

Running Beautify on a longer set of code like this to find out where the indentation changes are can drive you insane.  Instead, look at the results from running an action diagram:

┌-------IF llSomeExpr

|       ┌-------IF TestSomeCondition

|       |               * Do all sorts of code here.

|       |               * Got some other code here.

|       |       ┌-------IF SomeThingWorks

|       |       |               * Do lots of stuff here.

|       |       └-------ENDIF

|       ├-------ELSE

|       |               * All the above stuff doesn't happen

|       |               * I want to do something completely different

|       |               * Blah

|       |               * Blah

|       |               * Blah

|       └-------ENDIF

The first thing you’ll notice is that the outermost IF isn’t matched to anything.  That tells us there’s a problem somewhere within that IF statement.  Look a little closer and you’ll see that the DO WHILE .T. is matched with a ELSE and ENDIF statement.  You know that’s not right, so the problem has to be within that DO WHILE statement.  Keep working your way inward, looking for commands that are mismatched until you come to the IF TestSomeCondition command that’s matched with an ELSE and an OTHERWISE statement.  That’s the most deeply nested code with a problem in this example.  Now that you’ve narrowed it down, you can find where the problem is.  In this particular case, I had an ELSE instead of an ENDIF on line 13.

A note on Action Diagrams: Action Diagrams are one of the output selections of the Documentation Wizard, which can be found under the Wizards submenu on the Tools menu.  To view an Action Diagram, use MODIFY COMMAND from the VFP command window, and change the display to FoxFont or else the lines won’t show correctly.

“Syntax error”

When this error occurs on a command that’s a single line, it’s usually pretty obvious what the problem is when you look at the code.  But say you get a “Syntax error” on a SQL Select statement like this one:

SELECT ToolMast.Name as ToolDesc, ;

       goApp.BuildFullCategory(ToolMast.p_ToolCat) as Category, ;

       ToolMast.BergNo, ;


 ToolMast.SerialNo, ;

       IIF(ToolCat.IsNumbered=.T.,ToolPHst.PurchDate,{}) as PurchDate,;

 
 NVL(brgvendl.Name,SPACE(50)) as VendName, ;

       ToolCat.IsNumbered, ;


 goApp.GetCurrentLoc("General",ToolMast.Key)+' - '+ ;

            goApp.GetCurrentLoc("Specific",ToolMast.Key) as Location, ;

 
 goApp.GetPurchPrice(ToolMast.Key) as PurchPrice, ;

 
 SPACE(100) as SortFld, ;

       SPACE(100) as SecSort ;

   FROM ToolMast ;


INNER JOIN ToolCat ;



ON ToolMast.p_ToolCat = ToolCat.Key ;


LEFT OUTER JOIN ToolPHst ;



ON ToolMast.Key = ToolPHst.p_ToolMast ;


LEFT OUTER JOIN brgvendl ;



ON ToolPHst.p_brgvendl = brgvendl.Key ;

   INTO CURSOR RptData ;

   ORDER BY &lcSortFld readwrite

How do you figure out where the syntax error is?  I don’t know about you, but I find my most common mistake is either to leave out a comma after a field in the field list, or to put in a comma after the last field in the field list.  But sometimes, as in the case above, that isn’t the problem.  What you need to do in this situation is to start taking the statement apart.  

My first suspicion is any UDFs or methods called.  Remove those items from field list.  If the problem still occurs,  next look for macros.  In the case above, I hadn’t defined the lcSortFld I’m using in the ORDER BY clause.  That caused the syntax error.  If the problem had persisted after removing the macro, my next step would have been to start removing joins one at a time, plus any fields in the field list that rely on the join I’m removing.  At some point the error will disappear, and that last piece of code you removed is most likely the problem.

“The name of the object file for XXX is already used by another program in the project” (2001)

Example of this error: Modify Project BadClassCall and try to build an executable.  You must be running VFP 7 or earlier to demonstrate the error.

First the good news.  This error won’t occur in VFP 8 in the situation described below.  In case you get this error in VFP 7, read on.

This error occurs when you’re trying to create an EXE for your project, and is accompanied by an error in your ERR file saying that a class you’re calling is undefined.  What’s more, you’ll see the class that VFP is calling “undefined” show up multiple times in your project!  The problem here is that you are defining a class in a PRG that is inheriting from a class in another PRG or VCX, and you are specifying the full path to that other class in the DEFINE CLASS statement.  For example, 

DEFINE CLASS MyTest as cusTest OF c:\demo\save\customclasses.prg 

ENDDEFINE

will cause this error.  

The way to fix this problem is to SET PROCEDURE TO the appropriate PRG, and not reference the class at all in your DEFINE CLASS statement.  An example would be:

SET PROCEDURE TO C:\demo\save\customclasses.prg ADDITIVE

DEFINE CLASS MyTest as CusTest

ENDDEFINE

You would see the same problem if I had used a VCX instead of a PRG for the CustomClasses library.

“Error loading file - record number n. "object" <or one of its members>” (1881)

There are a number of variations on this error.  You’ll get this error while trying to run a form.  Often, the only way to figure out what’s going on is to USE the SCX or VCX as a table (after all, it IS just a DBF with a different extension!).  Go to the record number referenced in the error.  Usually, you’ll find that’s a cursor that you’ve removed from the system, or some other obvious problem.  In these cases, deleting the offending record is all you need to do.  I have heard of having the problem when you don’t even have anything in the data environment, but that seems to be fairly rare.  If this does happen to you, you may be stuck with having to recreate your form from scratch.

Some more specific examples follow.

“… ControlSource: Variable XXX is not found...”

Example of this error: Do Form ErrorLoadingFile

One of the most typical situation is where you’ve changed the structure of an underlying cursor.

Figure 2.  Example 1 of “Error loading…” error while running a form.

With a message like the one above, you need to pay attention to the object referenced.  In this case, record 5 is a grid object on the form called grdDemoview.  Going to that record in the SCX wouldn’t help in this case.  Note that the error says “ControlSource: Variable ‘DUMMY’ is not found”.  This indicates that one of the columns in the grid is using a ControlSource called “DUMMY”.  In this case, I was using a view as the RecordSource for this grid, and I’ve removed the field “DUMMY” from the view. 

When you do change structure this way, it’s a good idea to search your project to find any references to the field name you’ve changed or removed.  VFP 8 has a new Code References tool available from the Tool menu pad that will find these references for you.  If you don’t have VFP 8, there’s a free tool called PJXSearch available from www.stevedingle.com that will search your project for a string.

“…Error loading the data environment.  Table is in use”

Example of this error: Run the program ErrorLoadingDE

Another very common error is when you have a table opened exclusively.

Figure 3.  Example 2 of “Error loading…” error while running a form.

Unfortunately, the message doesn’t tell you which table is opened exclusively.  Quite often it’s something that you were doing in the Command Window, so it’s the default datasession.

“…Parent: Cannot add XXX.  Class definition is cyclical”

Example of this error: Run the program CyclicClass

Here’s one you’re not as likely to see, but can be hard to figure out if you ever do.

Figure 4.  Example 3 of “Error loading…” error while running a form.

This error occurs when you are trying to instantiate a VCX-based form which contains nested container objects inheriting from the same parent.  It doesn’t matter if the form class is in the same VCX with the container class or not.  The trick to solving this one is to change the parent class for one of the containers.  This new subclass doesn’t even have to have any changes from the parent.

"Fatal error X while attempting to report error X" (***)

I like to quote Jim Booth when I explain this error.  He says that VFP is saying, “I’m choking on a chicken bone, but I can’t tell you because I’m choking on a chicken bone!”  In essence VFP has completely lost it!  It’s best just to shut down VFP and start again.

I have seen this error occur repeatedly in a case where a class defined in a PRG used the SYS(2335,0) to turn on unattended server mode. 

Figure 5.  Fatal error X while attempting to report error X.

The SYS(2335) command really only applies when you’re running the program as a COM object, and in the case where it caused a problem, the program wasn’t running as a COM object.  Normally, this command would just cause your program to ignore any commands, such as MESSAGEBOX(), that would display something.  In this case, it caused the “Error X while attempting to report error X”, but only when run as an EXE from the runtime.  (In other words, it didn’t fail if you ran the EXE from the command window.)

 “Cannot update the cursor” (111)

Example of error: Modify the project ReadOnly.  When you try to rebuild all files in the project, you will get this error.

First, the good news.  In VFP 8, is supposed to tell you which file cannot be updated, however, as of this writing, the file name in the message is missing.  It’s expected that this will be corrected before the product ships.

You’ll get this error if you’re doing anything that tries to write to a read-only file.  Often this can be the result of copying files from a CD to your hard drive.  Remember, everything copied from a CD is read-only and must be changed.  Just use Windows Explorer to select all the files that need their read-only flag removed, then right-click and select Properties.  You’ll get a dialog like that shown in figure 6.

Figure 6. Removing the read-only flag from files.

Click the Apply button and you’re done!

If you happen to get a “Cannot update the cursor” error when trying to build a project, it can be a real headache trying to track down which file is causing the problem.  Ted Roche has solved this problem with a program that displays a Wait Window before compiling each program in your project.  I’ve further enhanced Ted’s program by logging all the problems into a cursor that is then displayed in a browse window.  The only requirement is that you use VFP 7 or later because it uses the Readwrite clause.  This program is included with the sample code as WhichOne.prg.
"Internal consistency error"/C0000005 (1000/?)

Example of C0000005 error: Run the program C0000005

Both of these errors are similar and are Visual FoxPro’s version of a GPF.  Craig Berntson has some information about “Internal consistency error” on his web site (www.craigberntson.com).   I don’t recall ever hearing about C0000005 before VFP 6 was released.  In versions of VFP 6 before SP3, you’d get this error at the drop of a hat.  Most of the problems were fixed with SP3, and I wouldn’t advise running any version of VFP 6 before that.

Both errors can be tough to track down.  The best place to start is with the resource file.  If possible, SET RESOURCE OFF and see if the problem persists.  If you need your resource file to run, try and recreate it.

There is also a lot of information about C0000005 errors on the VFP wiki (http://fox.wikis.com) under the topics of C0000005ExError and ManualGarbageCollection.

For the rest of this discussion, I’ll focus on C0000005 errors.  There’s way too much about C0000005 errors for me to cover in this paper, but I’ll go over some of the most common sources of the error.

C0000005 can be broken down into two situations—errors that occur during your development and testing, and errors that occur during normal operation of your app.  I’ve found that errors that occur during normal operation are usually data related, whereas errors during development are programming problems.  Both are discussed below.

Data Problems

Besides the resource file, another source of C0000005 errors can be data corruption.  The first thing to do is to delete all your index tags and recreate them.  DO NOT USE REINDEX!  The Reindex command will only perpetuate any problems with index tags.  If the problems still persist, pin down which (if any) of your tables is causing this problem.  Open each table and use the COPY TO command to copy it to a temp file.  If the C0000005 error is due to corrupted data, you’ll get the error when copying the file.

Once you’ve determined which file is causing the problem, use a binary search to find the record with the problem.  First try and copy the first half of the file (using FOR RECNO() < some number in your COPY command).  If you didn’t get an error, go to the first record you didn’t copy and COPY REST to the temp file.  Next, take whichever half had the problem, and divide that in half, using a COPY TO command on each.  Continue this process until you’ve narrowed down which record is corrupted (assuming it’s only one record) and record as much data as you can from that record, then delete it and recreate it.  You probably won’t be able to pack the table with that record in there, but you can COPY TO a file FOR NOT DELETED(), Zap the original table, and APPEND the records back from the temp table.

Programming Problems

If you’ve determined that the error is not being caused by a bad resource file or data corruption, but the error is reproducible and always happens in the same place, it could be due to a dangling object reference in your code or some other programming error.  Pin the problem down to a line of code or a section of code.  Once you find it, if you don’t see anything wrong in the code, try and reproduce it in a small program.  If you can reproduce it, you may need to use a different technique to accomplish what you’re trying to do.

Case 1:

Another possible cause of a C0000005 error is the use of the  TEXTMERGE command.  Under some situations, VFP won’t add a terminating CHR(0) to the string.  If that string also occurs at the end of an allocated memory block, VFP will cause a buffer overrun, which results in a C0000005 error.  You can get around this problem by adding a CHR(0) to any TEXTMERGE strings you create.  The CHR(0) is only needed for the TEXTMERGE function, so good practice says you should immediately remove it
:

CHRTRAN(TEXTMERGE(lcString+CHR(0)), CHR(0), “”)

Case 2:

Recently I had a situation where I was consistently getting C0000005 errors in a form that called a report that had a lot of code in the report’s Data Environment Init method, but only when tracing through the code.  The error happened every time I tried to trace through the code, but not always in the same place.  Packing both the SCX and the FRX stopped the C0000005 errors!  This just proves how difficult this error can be to explain.

Case 3:

In another case, I saw the dialog in figure 7.

Figure 7.  Example of C0000005 error.

In this particular case, the line number referenced was an ENDSCAN command.  I had just been working on the code in this loop, so I took a very close look at the changes I had made.  The original code had been changing the values of an existing record, but I had just changed the code to add a new record instead.  The record I added was in a table being used in the SCAN FOR…ENDSCAN loop.  This is not a good practice.  Although it’s not guaranteed that doing this will cause a C0000005 error, it did in this case.  I needed to add new records to this table though, so I opened the same table in another work area with table buffering.  I performed a TABLEUPDATE after the ENDSCAN so as not to add any records to the table controlling the loop.  This fixed the problem, but I haven’t been able to reproduce it since.

Case 4:

Sometimes, a simple recursive call can cause a C0000005 error.  Here’s one that will reliably produce an error:

X = “EVAL(X)”

? EVAL(X)

This kind of thing isn’t something you’d want to do, but could easily end up doing by accident.  Again, you need to try to isolate the problem to a specific line of code, and then the cause of the problem is often pretty obvious.

If all else fails, you may need to start over, which will often fix the problem.  If none of these solutions work, there’s some slim chance that the error may be memory related. 

"Not enough memory for file map" (1150)

This error may actually be the result of too much memory!  It seems to crop up on systems with 1 gig of memory or more.  The way to solve this is to direct VFP on how much memory to allocate with the SYS(3050) function.  Steve Dingle was having problems with this error, and has come up with a formula for setting SYS(3050).

#DEFINE  diMEM_MAX_FOREGRD       256  && The max memory to allocate

#DEFINE  diMEM_FOREGRD_PRECENT   .90  && Percent of memory to allocate

 LOCAL liNewMemory,;

       liAvailableMemory

 m.liAvailableMemory =  ( VAL( SYS(1001) ) / 4 ) / (1024^2) 

 m.liNewMemory = INT( m.liAvailableMemory * diMEM_FOREGRD_PRECENT )

 IF m.liNewMemory > diMEM_MAX_FOREGRD

    m.liNewMemory = diMEM_MAX_FOREGRD

 ENDIF

 m.liNewMemory =  m.liNewMemory * (1024^2)

 =SYS(3050, 1, m.liNewMemory )  && Set foreground memory

 =SYS(3050, 2, m.liNewMemory )  && Set background memory

This code allocates 90% of memory to foreground and background buffer memory.  You may want to play around with this number.  You shouldn’t need to increase the maximum above 256 megs.  You also probably want to make these values configurable instead of hard coding them.  Then, if they need tweaking, you don’t have to change your code.  Steve also notes that playing with SYS(3050) can affect your app in a negative way so you should store an “Override” option somewhere, like in an INI file.  Then, check the INI file (or whatever) first for a value and if appropriate, use the formula.

There are some additional observations from Christof Lange that help to explain things further:

1. You can’t really find the optimum setting for SYS(3050) because it will vary with the amount of physical memory available. That changes depending on what else the machine is running.  If SYS(3050) is too low, it’s possible that VFP will start creating TMP files on disk, even though there’s plenty of memory available.  If you set it too high, Windows will use paged memory, which results in a lot of disk swapping.  If your TMP files are stored on a different disk from the one Windows uses for page swapping, you’re better off to err on the side of using the faster disk.

2. Any memory you load into the cache must be unloaded.  Although unloading is faster than loading, it still takes time. The cache is unloaded when you deactivate the VFP application and the background cache is significantly smaller than the foreground cache.

3. There's an upper limit for SYS(3050) which is most noticeable with 1 GB or more of RAM.  This problem is probably caused by the automatic determination of the buffer size. It seems that SYS(3050) is doing a file mapping which requires non-fragmented virtual memory and is limited to 2 GB for all file mappings in a process. If memory is already fragmented and/or more memory is allocated for other means, Windows might not have enough memory to create a file mapping of the calculated size or it takes away all space for further file mappings.

If you have a process that you know requires a lot of memory, you may want to do some garbage collection at that point. Executing SYS(1104) - which performs a garbage collection - does take some time. Hence, you shouldn't call it much more than necessary. Typically, you would place it before or after operations that require lots of memory like SELECT statements or batch updates.

“OLE IDispatch exception code "number" from "name": "name"” (1428)

OLE errors originate from outside of VFP, so you can’t deal with them in the same way you would with other errors.  All the online help tells you to do is refer to the documentation for the automation object that caused the error.  Sometimes, that can be your own EXE or DLL that you’re running as a COM server, so there won’t be any documentation for this situation..  The case of the “Catastrophic failure” error shown in Figure 8 was sent to me by a colleague who was calling one of his multi-threaded DLLs from another.  The frustrating thing about this is he couldn’t figure out how to reproduce it.

Figure 8. Example of OLE error.

OLE errors involve some decoding, although frequently the details of the error give you enough information to get an idea of what went wrong.  

The important piece of information is the error code, which is always shown in hexadecimal.  If you happen to have the winerror.h file handy (which comes with the Win32 SDK), you can find a pretty comprehensive list of possible error codes.  Most likely, though, the message it lists is the same as what was displayed in the error message.

There are also some additional resources you can turn to.  First, there’s an OLE help file available from Microsoft.  This file is referenced in Knowledge Base article Q132150, “OLE Knowledge Base Available as Help File”.  In this OLE Knowledge Base under article Q122957, you’ll find instructions for downloading something called DECODE32.EXE.  Although the information in this article is a bit dated, I was still able to download the DECODE32 tools from the FTP site.  Unfortunately, unless you know C, you’re not going to be able to make much use of it.

In any case, it helps to know a bit about the structure of the OLE error code.

Figure 9. OLE error code layout

S is the severity level, 0 indicating success, 1 indicating failure. In other words, real errors all have negative values.

Fields R, C, N, and r are all reserved.

Facility is the facility code: currently there are eleven facility codes defined.

	Value
	Facility

	0
	FACILITY_NULL

	1
	FACILITY_RPC

	2
	FACILITY_DISPATCH

	3
	FACILITY_STORAGE

	4
	FACILITY_ITF

	7
	FACILITY_WIN32

	8
	FACILITY_WINDOWS

	9
	FACILITY_SSPI

	10
	FACILITY_CONTROL

	11
	FACILITY_CERT

	12
	FACILITY_INTERNET


Figure 10. Defined facility codes.
Code is the facility's status code, a 16-bit value.

It’s the 16-bit status code that gives you the specific code for the error.

